



ELSEVIER

Available online at [www.sciencedirect.com](http://www.sciencedirect.com)

ScienceDirect

Electronic Notes in  
Theoretical Computer  
Science

Electronic Notes in Theoretical Computer Science 212 (2008) 225–239

[www.elsevier.com/locate/entcs](http://www.elsevier.com/locate/entcs)

# Building Certified Static Analysers by Modular Construction of Well-founded Lattices

David Pichardie

*INRIA Rennes - Bretagne Atlantique, France*

---

## Abstract

This paper presents fixpoint calculations on lattice structures as example of highly modular programming in a dependently typed functional language. We propose a library of Coq module functors for constructing complex lattices using efficient data structures. The lattice signature contains a well-foundedness proof obligation which ensures termination of generic fixpoint iteration algorithms. With this library, complex well-foundedness proofs can hence be constructed in a functorial fashion. This paper demonstrates the ability of the recent Coq module system in manipulating algebraic structures and extracting efficient Ocaml implementations from them. The second contribution of this work is a generic result, based on the constructive notion of accessibility predicate, about preservation of accessibility properties when combining relations.

*Keywords:* Proof assistant, Constructive proofs, Static analysis.

---

## 1 Introduction

Static program analyses rely on fixpoint computations on lattice structures to solve data flows equations. The basic algorithms are relatively simple, but lattice structures can be complex when dealing with realistic programming languages. Termination of these computations relies on specific properties of the lattice structures, as for example the condition that all ascending chains are eventually stationary. In this work, we aim at increasing confidence in static analysers by using the proof-as-programs paradigm: from a machine-checked correctness proof of an analysis, we extract a certified analyser. We use the extraction mechanism of the Coq proof assistant to extract Ocaml programs from constructive proofs. In earlier work, we presented a lattice library which allows the construction of complex lattices in a modular fashion [4]. It was shown how this library was used to construct large termination proofs based on the ascending chain condition. This paper presents a new version of this library, based now on the more general termination criteria of widening.

We first present in Section 2 the module signature that models the kind of lattice we want to build. In Section 3 we motivate this library with a challenging example of lattice to be built in Coq. Sections 4 and 5 then present various lattice functors proposed in the library. Section 4 discusses binary functors, in particular the product functor. Section 5 deals with a functor of functions with various possible implementations. Conclusions are given in Section 6.

We expect the reader to be familiar with the ML module system. The whole Coq development is available on-line<sup>1</sup>.

## Related work

This paper is a descendent of the work of Jones [10] where a modular construction of finite lattices was proposed in the Haskell programming language using type classes. Our lattice signatures are not restricted to ML function types but they are also equipped with a specification. This is a consequence of the expressiveness gap existing between the Haskell and the Coq type systems.

In earlier work, we already introduced the lattice library [4]. However, we mainly discussed the semantic proofs required for *certified analyses*. Only ascending chain conditions proofs were studied and few details were given about their constructions. The current paper proposes several improvements:

- Mechanical proofs about fixpoint iteration using widenings has never been reported before. Other existing works only deal with ascending chain condition [11,2,5,3]. Widening operators require more complex termination proofs.
- In particular, new theoretical results about accessibility predicates are necessary to handle product of widening operators in the constructive logic of Coq.
- We propose a modular notion of functions (see Section 5) which allows to construct termination proofs without relying on the actual implementation chosen. Previous proofs were specific to one implementation, and as a consequence it was very difficult to adapt them to new function implementations.

The technical contribution of this paper deals with the modular construction of large proof terminations in a proof assistant. Proving termination of static analysers is sometimes considered as useless because we only need to check the result of the analyser, if it terminates. Nevertheless, bugs concerning termination of fixpoint iteration are difficult to debug: when do you stop the analyser ? Because of their non-monotonous nature, widening operators break human intuition and sometimes leads to invalid termination proofs (as noticed by Antoine Miné [13] as regards [16]).

Few detailed constructive proofs about accessibility properties have been published. The reference in this field is the work of Paulson [15] where general rules to preserve accessibility properties are given. Many of our proofs depend on these rules, however the notion of widening operator required further extensions. As far as we know the result proved in Theorem 4.6 is new.

<sup>1</sup> <http://www.irisa.fr/lande/pichardie/lattice/>

## 2 Module signatures for lattices

This work is based on two algebraic structures: *partially ordered sets* (posets) and *lattices* (see [8] for standard definitions). To be precise we consider a more general notion than posets because the posets  $(A, \equiv, \sqsubseteq)$  we consider in this paper are in fact composed of a set  $A$  and a pre-order  $\sqsubseteq$ .  $\equiv$  is the associated equivalence relation.

In Coq, the corresponding definitions are given as module signatures (see Figure 1). The `Poset` signature reads as follow: a module of type (or signature) `Poset` must at least contain a type  $\tau$  (to model elements in the posets) and two relations `eq` and `order`. It must also contain proofs that `eq` (resp `order`) is an equivalence relation (resp. a partial order). These required proofs are represented with the keyword **Axiom**. At last, the two relations `eq` and `order` must come with a computable test function `eq_dec` and `order_dec`. The type of the operator `eq_dec` is a dependent type that expresses the following: for any  $x$  and  $y$  of type  $\tau$ , the function must return a boolean such that, if the boolean is true,  $x$  and  $y$  are equivalent, otherwise if it is false, they are not.

The `Lattice` signature includes all elements of the `Poset` signature with the command **Include** `Poset`<sup>2</sup>. A first consequence of these signature definitions is that the statement “every lattice is a poset” is free in Coq: a module satisfying the `Lattice` signature, satisfies the `Poset` signature too.

We will need a further property to be able to compute over-approximation of fixpoints in such structures. In our previous work [4] we considered the ascending chain condition but in this work we are interested in more general criterion: the existence of a *widening operator*.

The standard fixpoint iteration *à la* Kleene may require an important number of iterations before convergence. Moreover, some lattices used in static analysis do not respect the ascending chain condition (like the lattice of intervals used in Section 3). The solution proposed by Cousot and Cousot [7] is a fixpoint approximation by a post fixpoint. Such a post fixpoint is computed with an algorithm of the form  $x_0 = \perp$ , and  $\forall n, x_{n+1} = x_n \nabla f(x_n)$  with  $\nabla$  a binary operator on  $A$  which “extrapolates” its two arguments. The computed sequence should be increasing (property ensured if  $\nabla$  satisfies  $\forall x, y \in A, x \sqsubseteq x \nabla y$ ) and should over-approximate the classical iteration:  $f^n(\perp) \sqsubseteq x_n$  (property ensured if  $\nabla$  satisfies  $\forall x, y \in A, y \sqsubseteq x \nabla y$ ). A last condition ensures the computation convergence: after a finite number of steps, we must reach a post fixpoint. The criterion proposed in the literature is generally “for all increasing chains  $x_0 \sqsubseteq x_1 \sqsubseteq \dots \sqsubseteq x_n \sqsubseteq \dots$ , the chain  $y_0 = x_0, y_{n+1} = y_n \nabla x_{n+1}$  eventually reaches a rank  $k$  with  $y_k \equiv y_{k+1}$ ”.

In order to implement this algorithm in Coq, we will work with a definition which is better adapted to constructive proofs. This definition will be based on the notion of *accessibility* and of *noetherian*<sup>3</sup> relation [1].

**Definition 2.1 (Accessibility)** Given a relation  $\prec$  on a set  $A$ , the set  $\text{Acc}_\prec$  of

<sup>2</sup> This command is not currently available in the Coq system. It should be replaced by the complete list of elements found in the module.

<sup>3</sup> The Coq library uses the inappropriate name of well-founded relation.

```

Module Type Poset.
  Parameter t : Set.

  Parameter eq : t → t → Prop.
  Axiom eq_refl : ∀ x : t, eq x x.
  Axiom eq_sym : ∀ x y : t, eq x y → eq y x.
  Axiom eq_trans : ∀ x y z : t, eq x y → eq y z → eq x z.
  Parameter eq_dec : ∀ x y : t, {eq x y}+{¬ eq x y}.

  Parameter order : t → t → Prop.
  Axiom order_refl : ∀ x y : t, eq x y → order x y.
  Axiom order_antisym : ∀ x y : t, order x y → order y x → eq x y.
  Axiom order_trans : ∀ x y z : t, order x y → order y z → order x z.
  Parameter order_dec : ∀ x y : t, {order x y}+{¬ order x y}.
End Poset.

Module Type Lattice.
  Include Poset.

  Parameter join : t → t → t.
  Axiom join_bound1 : ∀ x y : t, order x (join x y).
  Axiom join_bound2 : ∀ x y : t, order y (join x y).
  Axiom join_least_upper_bound :
    ∀ x y z : t, order x z → order y z → order (join x y) z.

  Parameter meet : t → t → t.
  Axiom meet_bound1 : ∀ x y : t, order (meet x y) x.
  Axiom meet_bound2 : ∀ x y : t, order (meet x y) y.
  Axiom meet_greatest_lower_bound :
    ∀ x y z : t, order z x → order z y → order z (meet x y).

  Parameter bottom : t.
  Axiom bottom_is_bottom : ∀ x : t, order bottom x.
End Lattice.

```

Fig. 1. The lattice signature

*accessibles* elements with respect to  $\prec$  are inductively defined as

$$\frac{\forall y \in A, y \prec x \Rightarrow y \in \text{Acc}_{\prec}}{x \in \text{Acc}_{\prec}}$$

**Definition 2.2 (Noetherian relation)** A relation  $\prec$  on a set  $A$  is *noetherian* if all elements in  $A$  is accessible with respect to  $\prec$ .

Intuitively, an element is accessible with respect to a relation  $\prec$  if it is not the starting point of any infinite  $\prec$ -decreasing chain. A trivial example of accessible element is an element without predecessor.

In order to express this widening criterion with the accessibility notion, we need to define a relation where infinite chains will be prohibited. Such a relation is defined by  $(x_1, y_1) \prec_{\nabla} (x_2, y_2)$  iff  $x_2 \sqsubseteq x_1 \wedge y_1 \equiv y_2 \nabla x_1 \wedge y_1 \not\equiv y_2$ . Then, the following equivalence holds

$$\left( \begin{array}{l} \text{there exists a chain } x_0 \sqsubseteq \dots \sqsubseteq x_{n+1} \sqsubseteq \dots \\ \text{with } y_0 = x_0, \text{ and } \forall n, y_{n+1} = y_n \nabla x_n \\ \text{satisfying } \forall k, y_k \not\equiv y_{k+1} \end{array} \right) \iff \left( \begin{array}{l} \text{there exists a sequence } ((x_k, y_k))_{k \in \mathbb{N}} \\ \text{satisfying } x_0 = y_0 \\ \text{and } \forall k, (x_{k+1}, y_{k+1}) \prec_{\nabla} (x_k, y_k) \end{array} \right)$$

The classical criterion found in the literature can hence be formulated under the form

$$\forall x \in A, (x, x) \in \text{Acc}_{\prec_{\nabla}}$$

Note that we do not require all elements to be accessible, only those of the form  $(x, x)$  because they are potential starting points for iteration with widening.

Finally, these properties are collected in the `PosetWiden` interface given in Figure 2. The properties `widen_eq1` and `widen_eq2` ensure that  $\nabla$  respects the equivalence  $\equiv$  taken on  $A$ . The definition of the signature `LatticeWiden` (lattice with a widening operator) is expressed in a similar way.

Figure 3 gives the construction of the associated generic post fixpoint solver. This module is a functor that takes in argument a module of type `LatticeWiden` and build an operator `pfp_widen` that computes post fixpoints. It is expressed in the type of the operator as follow: given a function  $f$ , if  $f$  is monotone then the function returns an element  $x$  in the lattice that is a post fixpoint.

```

Module Type PosetWiden.
  Include Poset.

  Parameter widen : t → t → t.

  Parameter widen_bound1 : ∀ x y : t, order x (widen x y).
  Parameter widen_bound2 : ∀ x y : t, order y (widen x y).
  Parameter widen_eq1 : ∀ x y z : t, eq x y → eq (widen x z) (widen y z).
  Parameter widen_eq2 : ∀ x y z : t, eq x y → eq (widen z x) (widen z y).

  Definition widen_rel : (t*t) → (t*t) → Prop := fun x y =>
    order (fst y) (fst x) ∧
    eq (snd x) (widen (snd y) (fst x)) ∧
    ¬ eq (snd y) (snd x).

  Parameter widen_acc_property : ∀ x : t, Acc widen_rel (x,x).
End PosetWiden.

```

Fig. 2. The module signature for poset with a widening operator

```

Module PostFixPoint (L:LatticeWiden).
  Definition monotone f := ∀ x y, L.order x y → L.order (f x) (f y).
  Definition pfp_widen f : monotone f → { x:L.t | L.order (f x) x } :=
    (* ... omitted ... *)
End PostFixPoint.

```

Fig. 3. Postfixpoint computation

### 3 A challenging example

When formalizing analyses for realistic programming language, the underlying lattice may be complex, even for analyses of middle precision. We give here an example of such lattice in order to motivate and illustrate our lattice library.

The aim of this lattice is to abstract the memory of a Java virtual machine with a context-sensitive interval abstraction for numerical values and context-sensitive class abstraction for references. Because in Java, values are numerics or references it is natural to abstract them with a sum of lattice, here the sum of the set of class name and the interval domain [7].

$$\text{Val} = \wp(\text{ClassName}) + \text{Interval}$$

The global structure of the lattice is then of the form:

$$St = L \times H$$

with  $L$  and  $H$  some function domains of the form:

$$L = \text{Context} \rightarrow ((\text{Val})^* \times (\text{Var} \rightarrow \text{Val})) \quad H = \text{ClassName} \rightarrow (\text{FieldName} \rightarrow \text{Val})$$

$\text{Var}$ ,  $\text{ClassName}$ ,  $\text{FieldName}$ ,  $\text{MethodName}$  and  $\text{ProgPoint}$  represent here the (finite) sets of variable name, class name, field name, method name and program points. All this set are encoded with integers on 32 bits. The set  $\text{Context}$  is composed of list of couples in  $\text{MethodName} \times \text{ProgPoint}$ . These lists have at most  $k$  elements and represent the last  $k$  call sites.

$$\text{Context} = (\text{MethodName} \times \text{ProgPoint})^{*\leq k}$$

$L$  is the flow-sensitive local abstraction of operand stack and local variables.  $H$  is the flow-insensitive abstraction of the heap.

The global domain  $St$  admit a lattice structure with a widening operator. Thanks to our lattice library it can be simply built by composition of functors. The construction is presented in Figure 4. For  $\text{Val}$  we use a functor `SumLiftLatticeWiden` that builds the disjoint sum of two lattices. We build  $H$  with the function functor presented in Section 5. The `MapLatticeWiden` functor allows to build function with a complex codomain (here  $\text{Context}$ ). Its utilisation (corresponding to line 8 to 11) will be explained in Section 5. The final lattice is built with the product functor `ProdLatticeWiden` presented in the next section.

```

1  Module Val := SumLiftLatticeWiden(IntervalLattice) (FiniteSetLatticeWiden) .
2
3  Module H := ArrayBinLatticeWiden(ArrayBinLatticeWiden(Val)) .
4
5  Module LocalVar := ArrayBinLatticeWiden Val .
6  Module Stack := ListLiftLatticeWiden Val .
7
8  Module N5. Definition val : nat := 5. End N5.
9  Module Context := ListFiniteSet(N5) (ProdFiniteSet(WordFiniteSet) (WordFiniteSet)) .
10 Module Map := FMapList.Make Context .
11 Module L := MapLatticeWiden(Context) (Map) (ProdLatticeWiden(Stack) (LocalVar)) .
12
13 Module GlobalState := ProdLatticeWiden(L) (H) .

```

Fig. 4. Construction of the global lattice in Coq

We will now present some of the functors introduced during this example. We will generally focus on the poset part (with widening) of the modules because the operators that are specific to lattice do not require any technical details.

## 4 Lattice functors

We propose three basic binary functors in our library: the product, the disjoint sum and the lifted sum. Due to lack of space, we will restrict our explanations in this paper to the product.

#### 4.1 Poset product

**Lemma 4.1 (Poset product)** *Given two posets  $(A, \equiv_A, \sqsubseteq_A)$  and  $(B, \equiv_B, \sqsubseteq_B)$ , the triplet  $(A \times B, \equiv_{A \times B}, \sqsubseteq_{A \times B})$  defined by  $\equiv_{A \times B} = \{((a_1, b_1), (a_2, b_2)) \mid a_1 \equiv_A a_2 \wedge b_1 \equiv_B b_2\}$  and  $\sqsubseteq_{A \times B} = \{((a_1, b_1), (a_2, b_2)) \mid a_1 \sqsubseteq_A a_2 \wedge b_1 \sqsubseteq_B b_2\}$  is a poset, called poset product.*

In Coq, Lemma 4.1 corresponds to a functor which takes two modules of signature `Poset` and returns a module respecting the `Poset` signature for the product structure.

#### 4.2 The poset-with-widening product

```

Module ProdPosetWiden (P1:PosetWiden) (P2:PosetWiden) : PosetWiden
with Definition t := (P1.t * P2.t)
with Definition eq := fun (x y : (P1.t * P2.t)) =>
  P1.eq (fst x) (fst y)  $\wedge$  P2.eq (snd x) (snd y)
with Definition order := fun (x y : (P1.t * P2.t)) =>
  P1.order (fst x) (fst y)  $\wedge$  P2.order (snd x) (snd y)
with Definition widen := fun (x y : (P1.t * P2.t)) =>
  match (x,y) with
    ((x1,x2),(y1,y2)) => (P1.widen x1 y1, P2.widen x2 y2)
  end.

Include ProdPoset (P1) (P2).

Definition widen (x y : t) :=
  match (x,y) with
    ((x1,x2),(y1,y2)) => (P1.widen x1 y1, P2.widen x2 y2)
  end.
...

Lemma widen_acc_property :  $\forall x:t, \text{Acc widen\_rel } (x,x)$ .
Proof. ... Qed.

End ProdPosetWiden.

```

Fig. 5. The poset-with-widening product functor

The construction of the poset-with-widening product functor is given in Figure 5. The interactive definition of this functor is made in three steps. We first give the functor signature with its base type `t`, its equivalence relation `eq`, its order relation `order` and the considered widening operator using the **with** notation. In a second step, we construct the definitions dealing with the poset part using the poset product functor `ProdPoset`. Note that in the expression `ProdPoset (P1) (P2)`, modules `P1` and `P2` are used as module of type `Poset`. The signature inclusion of `Poset` into `PosetWiden` allows this use without requiring any proof of coercion. This is a convenient feature when manipulating nested algebraic structures.

The last step concerns the new part of this functor: the proof that the widening operator satisfies its termination criterion. In our previous work [4] the termination criterion for the product of noetherian poset (*i.e.* that satisfy the ascending chain condition) was proved using a classical result about lexicographic products, but it is not possible for widening operators. Indeed, the key lemma to be established is:

**Lemma 4.2** *Given two posets  $(A, \equiv_A, \sqsubseteq_A)$  and  $(B, \equiv_B, \sqsubseteq_B)$ , two binary operators  $\nabla_A$  and  $\nabla_B$  on  $A$  and  $B$ , if  $\forall a \in A, (a, a) \in \text{Acc}_{\prec_{\nabla_A}}$  and  $\forall b \in B, (b, b) \in \text{Acc}_{\prec_{\nabla_B}}$  then the operator  $\nabla_{A \times B}$  defined by*

$$(a_1, b_1) \nabla_{A \times B} (a_2, b_2) = (a_1 \nabla_A a_2, b_1 \nabla_B b_2)$$

*satisfies  $\forall c \in A \times B, (c, c) \in \text{Acc}_{\prec_{\nabla_{A \times B}}}$ .*

This result is standard when proved in classical logic [7]. In constructive logic, it has never been proved before (as far as we know). It requires a technical proof to be directly established (because by example, it relies on pairs of pairs). We can make a more general proof and express the current problem as a particular case. The idea consists in expressing  $\prec_{\nabla_{A \times B}}$  as a lexicographic product between  $\prec_{\nabla_A}$  and  $\prec_{\nabla_B}$ . We then have to prove a result of the form

$$\forall a \in \text{Acc}_{\triangleleft_A}, \forall b \in \text{Acc}_{\triangleleft_B}, (a, b) \in \text{Acc}_{\triangleleft_{A \times B}^{\text{lex}}}$$

with  $\triangleleft_A$  playing the role of  $\prec_{\nabla_A}$  and  $\triangleleft_B$  the one of  $\prec_{\nabla_B}$ . However if  $\triangleleft_{A \times B}^{\text{lex}}$  denotes the standard lexicographic product of the two relations, the result is generally false:

**Lemma 4.3** *Given two relations  $\triangleleft_A$  and  $\triangleleft_B$  on sets  $A$  and  $B$ , if  $a \in \text{Acc}_{\triangleleft_A}$  and  $b \in \text{Acc}_{\triangleleft_B}$ , if there exist  $b' \in B$  such that  $b' \notin \text{Acc}_{\triangleleft_B}$  and  $a' \in A$  such that  $a' \triangleleft_A a$  then  $(a, b) \notin \text{Acc}_{\triangleleft_{A \times B}^{\text{lex}}}$ .*

**Proof.** Let us suppose that  $(a, b) \in \text{Acc}_{\triangleleft_{A \times B}^{\text{lex}}}$  and show that  $b' \in \text{Acc}_{\triangleleft_B}$  then holds. Because  $(a', b') \triangleleft_{A \times B}^{\text{lex}} (a, b)$ , we have  $(a', b') \in \text{Acc}_{\triangleleft_{A \times B}^{\text{lex}}}$ . But it implies  $b' \in \text{Acc}_{\triangleleft_B}$ . Indeed, if we consider the function  $f : B \rightarrow (A \times B)$  defined by  $f(x) = (a', x)$ , we have

$$\forall b_1, b_2 \in B, b_1 \triangleleft_B b_2 \Rightarrow f(b_1) \triangleleft_{A \times B}^{\text{lex}} f(b_2)$$

so we can apply the Lemma A.3 with  $f(b') = (a', b')$  to conclude.  $\square$

The problem here is that we can take any element  $b'$  to obtain a predecessor  $(a', b')$  of  $(a, b)$ . The case  $a_1 \triangleleft_A a_2$  in the definition of  $\triangleleft_{A \times B}^{\text{lex}}$  is hence too weak. We have to make restrictions on  $b_1$  and  $b_2$ . To this purpose, we introduce a relation  $\blacktriangleleft_B$  and propose a new product of the form

$$(a_1, b_1) \triangleleft_{A \times B}^{\text{lex}} (a_2, b_2) \iff (a_1 \triangleleft_A a_2 \text{ and } b_1 \blacktriangleleft_B b_2) \text{ or } (a_1 = a_2 \text{ and } b_1 \triangleleft_B b_2)$$

adding a constraint between  $\triangleleft_B$  and  $\blacktriangleleft_B$  to prevent having any  $b'$  as previously: if  $b_2 \blacktriangleleft_B b_1$  and  $b_1 \in \text{Acc}_{\triangleleft_B}$  then  $b_2$  should stay in  $\text{Acc}_{\triangleleft_B}$ . We will take a simpler sufficient condition (requiring no accessibility):

$$\forall b_1, b_2, b_3 \in B, b_1 \triangleleft_B b_2 \text{ and } b_2 \blacktriangleleft_B b_3 \text{ implies } b_1 \triangleleft^+ b_3$$

We can even propose a symmetric definition and encompass the case of  $\sqsubseteq_{A \times B}$  (where  $a_1 = a_2$  was replaced by  $a_1 \equiv_A a_2$ ) by introducing a relation  $\blacktriangleleft_A$  satisfying a similar property than  $\blacktriangleleft_B$ .



**Definition 4.4 (Extended lexicographic product)** Given two pairs of relations  $\triangleleft_A$  and  $\blacktriangleleft_A$  on a set  $A$ ,  $\triangleleft_B$  and  $\blacktriangleleft_B$  on  $B$ . The *extended lexicographic product* is the relation  $\blacktriangleleft^{\text{lex}(\triangleleft_A, \triangleleft_B, \blacktriangleleft_A, \blacktriangleleft_B)}$  defined on  $A \times B$  by

$$(a_1, b_1) \blacktriangleleft^{\text{lex}(\triangleleft_A, \triangleleft_B, \blacktriangleleft_A, \blacktriangleleft_B)} (a_2, b_2) \iff (a_1 \triangleleft_A a_2 \text{ and } b_1 \blacktriangleleft_B b_2) \text{ or } (a_1 \blacktriangleleft_A a_2 \text{ and } b_1 \triangleleft_B b_2)$$

with the following conditions

$$\forall a_1, a_2, a_3 \in A, a_1 \triangleleft_A a_2 \text{ and } a_2 \blacktriangleleft_A a_3 \text{ implies } a_1 \triangleleft_A^+ a_3 \quad (1)$$

$$\forall b_1, b_2, b_3 \in B, b_1 \triangleleft_B b_2 \text{ and } b_2 \blacktriangleleft_B b_3 \text{ implies } b_1 \triangleleft_B^+ b_3 \quad (2)$$

When the context will allow us to do it without ambiguity, we will use  $\blacktriangleleft$  to denote this relation.

**Example 4.5** The standard lexicographic product is a special case of  $\blacktriangleleft$ .

$$(a_1, b_1) \sqsupset_{A \times B} (a_2, b_2) \iff (a_1 \sqsupset_A a_2 \text{ and } b_1 \sqsupseteq b_2) \text{ or } (a_1 \equiv_A a_2 \text{ and } b_1 \sqsupset b_2)$$

and we have

$$\forall a_1, a_2, a_3 \in A, a_1 \sqsupset_A a_2 \text{ and } a_2 \equiv_A a_3 \text{ implies } a_1 \sqsupset_A a_3$$

and

$$\forall b_1, b_2, b_3 \in B, b_1 \sqsupset_B b_2 \text{ and } b_2 \sqsupseteq_B b_3 \text{ implies } b_1 \sqsupset_B^+ b_3$$

Then  $\sqsupset_{A \times B} = \blacktriangleleft^{\text{lex}(\sqsupset_A, \sqsupseteq_B, \equiv_A, \sqsupseteq_B)}$ .

**Theorem 4.6** If  $\triangleleft_A$ ,  $\triangleleft_B$ ,  $\blacktriangleleft_A$  and  $\blacktriangleleft_B$  satisfy the hypotheses of the previous definition, then for all  $a \in \text{Acc}_{\triangleleft_A}$  and  $b \in \text{Acc}_{\triangleleft_B}$ ,  $(a, b) \in \text{Acc}_{\blacktriangleleft}$ .

**Proof.** The form of this statement encourages to use a noetherian induction on the property

$$\forall a \in A, \mathcal{P}(a) := "\forall b \in \text{Acc}_{\triangleleft_B}, (a, b) \in \text{Acc}_{\blacktriangleleft}"$$

But the generated induction principle will be too weak because only usable for an element  $b \in \text{Acc}_{\triangleleft_B}$ .

We hence take a stronger goal by proving

$$\forall a \in \text{Acc}_{\triangleleft_A^+}, \forall b_1 \in \text{Acc}_{\triangleleft_B}, \forall b_2 \in B, b_2 \blacktriangleleft_B^* b_1, \Rightarrow (a, b_2) \in \text{Acc}_{\blacktriangleleft} \quad (3)$$

with  $\blacktriangleleft_B^*$  the reflexive transitive closure of  $\blacktriangleleft_B$ . This result is sufficient to establish our theorem because  $\blacktriangleleft_B^*$  is reflexive and because  $\text{Acc}_{\triangleleft_A} = \text{Acc}_{\triangleleft_A^+}$  (using Lemma A.4). To prove (3), we use a noetherian induction on  $a$  and  $\triangleleft_A^+$ . We consider an element  $a_1 \in \text{Acc}_{\triangleleft_A^+}$  such that

$$\begin{aligned} \forall a_2 \in A, a_2 \triangleleft_A^+ a_1 &\Rightarrow \\ \forall b_1 \in \text{Acc}_{\triangleleft_B}, \forall b_2 \in B, b_2 \blacktriangleleft_B^* b_1, &\Rightarrow (a_2, b_2) \in \text{Acc}_{\blacktriangleleft} \end{aligned} \quad (4)$$

and try to prove  $\forall b_1 \in \text{Acc}_{\triangleleft_B}, \forall b_2 \in B, b_2 \blacktriangleleft_B^* b_1, \Rightarrow (a_1, b_2) \in \text{Acc}_{\triangleleft}$ . Once time again a direct proof by induction on  $b_1$  will be unsuccessful without reinforcing first the current goal. We prove instead:

$$\begin{aligned} \forall b_1 \in \text{Acc}_{\triangleleft_B^+}, \\ \forall b_2 \in B, b_2 \blacktriangleleft_B^* b_1, \Rightarrow \\ \forall a_3 \in A, a_3 \blacktriangleleft_A^* a_1 \Rightarrow (a_3, b_2) \in \text{Acc}_{\triangleleft} \end{aligned} \quad (5)$$

This is a sufficient result because  $\text{Acc}_{\triangleleft_B} = \text{Acc}_{\triangleleft_B^+}$  and  $\blacktriangleleft_A^*$  is reflexive. We prove (5) by noetherian induction on  $b_1$ . Hence we take an element  $b_1 \in \text{Acc}_{\triangleleft_B^+}$  such that

$$\begin{aligned} \forall b_1 \in \text{Acc}_{\triangleleft_B^+}, \forall b_3 \in B, b_3 \triangleleft_B^+ b_1 \Rightarrow \\ \forall b_2 \in B, b_2 \blacktriangleleft_B^* b_3, \Rightarrow \\ \forall a_3 \in A, a_3 \blacktriangleleft_A^* a_1 \Rightarrow (a_3, b_2) \in \text{Acc}_{\triangleleft} \end{aligned} \quad (6)$$

and we look for a proof of

$$\forall b_2 \in B, b_2 \blacktriangleleft_B^* b_1, \Rightarrow \forall a_3 \in A, a_3 \blacktriangleleft_A^* a_1 \Rightarrow (a_3, b_2) \in \text{Acc}_{\triangleleft}$$

Let us suppose

$$b_1 \in \text{Acc}_{\triangleleft_B^+} \quad (7)$$

$$b_2 \blacktriangleleft_B^* b_1 \quad (8)$$

$$a_3 \blacktriangleleft_A^* a_1 \quad (9)$$

and prove  $(a_3, b_2) \in \text{Acc}_{\triangleleft}$ . We have to consider a predecessor  $(a_4, b_4)$

$$(a_4, b_4) \triangleleft_{\triangleleft} (a_3, b_2) \quad (10)$$

and we have to prove  $(a_4, b_4) \in \text{Acc}_{\triangleleft}$ . Using the definition of  $\triangleleft_{\triangleleft}$ , two cases results from hypothesis (10).

- Case 1 :

$$a_4 \triangleleft_A a_3 \quad (11)$$

$$b_4 \blacktriangleleft_B b_2 \quad (12)$$

We can use the induction hypothesis (4). Three conditions must be satisfied

- $a_4 \triangleleft_A a_1$  : true using hypotheses (1), (9) and (11).
- $b_1 \in \text{Acc}_{\triangleleft_B}$  : true using (7) and the general result  $\text{Acc}_{\triangleleft_B^+} = \text{Acc}_{\triangleleft_B}$
- $b_4 \blacktriangleleft_B^* b_1$  : true because of (12) and (8)

- Case 2 :

$$a_4 \blacktriangleleft_A a_3 \quad (13)$$

$$b_4 \blacktriangleleft_B b_2 \quad (14)$$

This time, we can use the induction hypothesis (6). Three conditions must be satisfied

- $b_4 \blacktriangleleft_A^+ b_1$  : true because of the hypotheses (2), (8) and (14)
- $b_4 \blacktriangleleft_B^* b_4$  : true by reflexivity of  $\blacktriangleleft_B^*$
- $a_4 \blacktriangleleft_A^* a_1$  : true using hypotheses (9) and (13)

Hence the main result is established.  $\square$

To prove Lemma 4.2, we only have to use a *measure function* (see Lemma A.3)  $f : (A \times B) \times (A \times B) \rightarrow (A \times A) \times (B \times B)$  defined by  $f((a_1, b_1), (a_2, b_2)) = ((a_1, a_2), (b_1, b_2))$  and considering the relation  $\prec_{\nabla_A \times B}$  on  $(A \times B) \times (A \times B)$  and  $\blacktriangleleft^{\text{lex}}(\prec_{\nabla_A}, \prec_{\nabla_B}, \preceq_{\nabla_A}, \preceq_{\nabla_B})$  on  $(A \times A) \times (B \times B)$  where  $\preceq_{\nabla_A}$  and  $\preceq_{\nabla_B}$  are defined by  $(x_1, y_1) \preceq_{\nabla_A} (x_2, y_2) \iff x_2 \sqsubseteq_A x_1 \wedge y_1 \equiv y_2$  and  $(x_1, y_1) \preceq_{\nabla_B} (x_2, y_2) \iff x_2 \sqsubseteq_B x_1 \wedge y_1 \equiv_B y_2$ . It is not difficult to verify that hypotheses of Theorem 4.6 are fulfilled and then conclude.

## 5 Lattices of functions

Another important functor concerns functions. Static analyses make heavy use of functions during their computations. Efficiency of the underlying data structures is hence crucial. However proof of termination properties on complex data structures can be hard. This section proposes an abstract notion of function implementation for which we prove termination properties. These proof can then be used for several efficient implementations. We now describe the functor which builds a poset with widening for functions.

First, we remark that implementing functions with the native functions of the chosen functional programming language is not a reasonable solution. It is better to use encoding as association lists, balanced trees, ... We will then prove the termination criterion of widening "for all function implementations".

### 5.1 Function implementation

The notion of function implementation is given in Figure 6. This signature handles

- a module <sup>4</sup>  $A$  with a signature `FiniteSet` (associated with the function domain). The `FiniteSet` signature is given in Figure 7. It represents set in bijection with parts  $\llbracket 0, \text{cardinal} - 1 \rrbracket$  of  $\mathbb{Z}$ . Our library proposes finite set functors (product, list of bounded length) and a base finite set module (binary number on 32 bits).
- a poset module  $B$  associated with codomain.

<sup>4</sup> Modules can handles modules. The corresponding signature element is then introduced by **Declare Module**.

```

Module Type Func_FiniteSet_PosetWiden.
Declare Module A : FiniteSet.
Declare Module B : PosetWiden.

Parameter t : Set.

Parameter get : t → A.t → B.t.

Definition eq : t → t → Prop := fun f1 f2 =>
  ∀ a1 a2 : A.t, A.eq a1 a2 → B.eq (get f1 a1) (get f2 a2).
Axiom eq_refl : ∀ x : t, eq x x.
Axiom eq_dec : ∀ x y : t, {eq x y}+{¬ eq x y}.

Definition order : t → t → Prop := fun f1 f2 =>
  ∀ a1 a2 : A.t, A.eq a1 a2 → B.order (get f1 a1) (get f2 a2).
Parameter order_dec : ∀ x y : t, {order x y}+{¬ order x y}.
End Func_FiniteSet_PosetWiden.

```

Fig. 6. Function implementation signature

- an abstract type  $t$  used to represent functions.
- a function `get` where `(get F a)` gives the image of  $a:A.t$  for the function associated with the element  $F:t$ .
- fixed equivalence (`eq`) and order (`order`) relation definitions with their test implementations (`eq_dec` and `order_dec`).
- the property `eq_refl` ensures that `get` is compatible with the equivalence relation  $A.eq$  taken on  $A.t$ .

```

Module Type FiniteSet.
Parameter t : Set.

Parameter eq, eq_dec [...]
Axiom eq_refl, eq_sym, eq_trans [...]

Parameter cardinal : Z.
Axiom cardinal_positive : cardinal > 0.

Parameter inject : t → Z.
Parameter nat2t : Z → t.
Axiom inject_bounded : ∀ x : t, 0 <= (inject x) < cardinal.
Axiom inject_nat2t : ∀ n : Z, 0 <= n < cardinal → inject (nat2t n) = n.
Axiom inject_injective : ∀ x y : t, inject x = inject y → eq x y.
Axiom inject_comp_eq : ∀ x y : t, eq x y → inject x = inject y.
End FiniteSet.

```

Fig. 7. FiniteSet signature

## 5.2 A widening operator on functions

Now for any function implementation we build a poset with a standard widening operator. For functions in  $A \rightarrow B$  this operator is defined as

$$\forall f_1, f_2 \in A \rightarrow B, \forall a \in A, (f_1 \nabla f_2)(a) = f_1(a) \nabla_A f_2(a)$$

The proof of the termination criterion relies on the Theorem 4.6 and the finiteness of the codomain.

### 5.3 Two efficient implementations

We propose two function implementations in our library. The first is a specific implementation for functions whose domain is a bounded binary integer (each integer denotes a position in a tree [14]). This kind of efficient implementation is heavily used in Leroy’s certified compiler [12]. The second implementation is based on an abstract implementation of Ocaml maps. We have adapted the Ocaml signature to Coq and proven that any map fulfils the `Func_FiniteSet_PosetWiden` signature. We currently propose a sorted list implementation and plan an implementation with balanced tree, both based on the previous formalisation done in [9]<sup>5</sup>. Maps can be built on any finite set. Finite sets can be constructed with the previously enumerated functors.

To conclude this section, we finish by commenting the example presented in Figure 4. `Context` is a module of type `FiniteSet` that is built with the functor `ListFiniteSet`. `N5` is a module that encapsulate the natural number 5. We hence bound our lists with at most 5 elements. `MapLatticeWiden` is a functor that take as argument a finite set (here `Context`), a map implementation (here `Map` built with sorted list) and a lattice with a widening operator. It builds the expected lattice and its widening operator.

## 6 Conclusion

We have presented a framework for programming fixpoint computations on lattice structures in a dependently typed functional language. In order to construct complex lattices, we propose a library of Coq module functors. We focused our explanations on the product and the function functor, but other functors are available in our Coq development.

The main contribution of this work deals with constructive proofs of termination properties. The termination criteria used with widening operators has required extensions of previously known results about accessibility predicates. Termination proofs are often very difficult to do in a proof assistant. This library shows the benefit of modular reasoning to handle such complex proofs. By composing the various functors that we propose, it is now possible to easily construct termination proofs for deep structures with efficient extracted data structures in Ocaml.

We have more recently extend our library to handle narrowing operators [7]. Again the technical difficulty relies in the functor product. It is interesting to notice that termination criterion of the narrowing operator is proved with the Theorem 4.6. It confirms that this theoretical result was a cornerstone for our work.

We imagine two extension for our library. The first one concerns the construction of base lattices, those which are used to instantiate lattice functors and construct bigger lattices. Some automation could be proposed to quickly construct finite lattices with their correctness proofs starting from a text description of their Hasse diagram. The second one concerns Galois connexion that could be constructed in

<sup>5</sup> Balanced trees are a keystone of the industrial-task Astre static analyser [6].

the same modular way.

## References

- [1] Peter Aczel. An introduction to inductive definitions. In J. Barwise, editor, *Handbook of Mathematical Logic*, pages 739–. North-Holland Publishing Company, 1977.
- [2] Gilles Barthe and Guillaume Dufay. A tool-assisted framework for certified bytecode verification. In *Proc. of the 7th International Conference on Fundamental Approaches to Software Engineering, (FASE'04)*, volume 2984 of *LNCS*, pages 99–113. Springer, 2004.
- [3] Yves Bertot, Benjamin Grégoire, and Xavier Leroy. A structured approach to proving compiler optimizations based on dataflow analysis. In *Proc. of the International Workshop on Types for Proofs and Programs (TYPES'04)*, volume 3839 of *LNCS*, pages 66–81. Springer, 2004.
- [4] David Cachera, Thomas Jensen, David Pichardie, and Vlad Rusu. Extracting a Data Flow Analyser in Constructive Logic. *Theoretical Computer Science*, 342(1):56–78, September 2005.
- [5] Solange Coupet-Grimal and William Delobel. A uniform and certified approach for two static analyses. In *Proc. of the International Workshop on Types for Proofs and Programs (TYPES'04)*, volume 3839 of *LNCS*, pages 115–137. Springer, 2004.
- [6] P. Cousot, R. Cousot, J. Feret, L. Mauborgne, A. Miné, D. Monniaux, and X. Rival. The ASTRÉE Analyser. In *Proc. of the 14th European Symposium on Programming Languages and Systems (ESOP'05)*, volume 3444 of *LNCS*, pages 21–30. Springer, 2005.
- [7] Patrick Cousot and Radhia Cousot. Systematic design of program analysis frameworks. In *Proceedings of the 6th Symposium on Principles of Programming Languages (POPL'79)*, pages 269–282. ACM Press, New York, 1979.
- [8] B.A. Davey and H.A. Priestley. *Introduction to Lattices and Order*. Cambridge University Press, 1990.
- [9] Jean-Christophe Filliâtre and Pierre Letouzey. Functors for Proofs and Programs. In *Proc. of the 13th European Symposium on Programming Languages and Systems (ESOP'04)*, number 2986 in *LNCS*, pages 370–384. Springer-Verlag, 2004.
- [10] Mark P. Jones. Computing with lattices: An application of type classes. *Journal of Functional Programming*, 2(4):475–503, October 1992.
- [11] Gerwin Klein and Tobias Nipkow. Verified Bytecode Verifiers. *Theoretical Computer Science*, 298(3):583–626, 2002.
- [12] Xavier Leroy. Formal certification of a compiler back-end, or: programming a compiler with a proof assistant. In *Proc. of 33th ACM Symposium on Principles of Programming Languages (POPL'06)*, pages 42–54. ACM Press, 2006.
- [13] Antoine Miné. The octagon abstract domain. In *AST 2001 in WCRE 2001*, IEEE, pages 310–319. IEEE CS Press, October 2001.
- [14] Chris Okasaki and Andrew Gill. Fast mergeable integer maps. In *Proc. of the ACM SIGPLAN Workshop on ML*, pages 77–86, 1998.
- [15] Lawrence C. Paulson. Constructing recursion operators in intuitionistic type theory. *Journal of Symbolic Computation*, 2(4):325–355, December 1986.
- [16] Ran Shaham, Elliot K. Kolodner, and Shmuel Sagiv. Automatic removal of array memory leaks in java. In *Proc. of the 9th International Conference on Compiler Construction (CC'00)*, LNCS, pages 50–66. Springer, 2000.

## A General results about accessibility predicates

The following results are proved by Paulson in [15] and belong to the Coq standard library.

**Lemma A.1** *For all relations  $R_1, R_2$  on a set  $A$ , if  $R_1 \subseteq R_2$  then  $\text{Acc}_{R_2} \subseteq \text{Acc}_{R_1}$ .*

**Lemma A.2** *For all relation  $R_B$  on a set  $B$ , for all function  $f : A \rightarrow B$  with  $A$  a set, the relation  $R_A$  defined on  $A$  by  $R_A = \{(a_1, a_2) \mid (f(a_1), f(a_2)) \in R_B\}$  satisfies  $\forall a \in A, f(a) \in \text{Acc}_{R_B} \Rightarrow a \in \text{Acc}_{R_A}$ .*

These two lemmas can be summarised in the following result

**Lemma A.3** *For all relations  $R_A$  on a set  $A$ ,  $R_B$  on a set  $B$ , for all function  $f : A \rightarrow B$  satisfying  $\forall (a_1, a_2) \in R_A, (f(a_1), f(a_2)) \in R_B$ , we have  $\forall a \in A, f(a) \in \text{Acc}_{R_B} \Rightarrow a \in \text{Acc}_{R_A}$ .*

$f$  is often called a *measure function*.

**Lemma A.4** *For all relation  $R$  on a set  $A$ ,  $\text{Acc}_R = \text{Acc}_{R^+}$  with  $R^+$  the transitive closure of  $R$ .*